Applications of DEs:
Spring problems

Prof. Joyner[1]

Ut tensio, sic vis[2].

- *Robert Hooke, 1678*

One of the ways DEs arise is by means of modeling physical phenomenon, such as spring equations. For these problems, consider a spring suspended from a ceiling. We shall consider three cases: (1) no mass is attached at the end of the spring, (2) a mass is attached and the system is in the rest position, (3) a mass is attached and the mass has been displaced fomr the rest position.



Figure 1: A spring at rest, without mass attached.



Figure 2: A spring at rest, with mass attached.



Figure 3: A spring in motion.

[2]"As the extension, so the force."

One can also align the springs left-to-right instead of top-to-bottom, without changing the discussion below.

Notation: Consider the first two situations above: (a) a spring at rest, without mass attached and (b) a spring at rest, with mass attached. The distance the mass pulls the spring down is sometimes called the "stretch", and denoted s. (A formula for s will be given later.)

Now place the mass in motion by imparting some initial velocity (tapping it upwards with a hammer, say, and start your timer). Consider the second two situations above: (a) a spring at rest, with mass attached and (b) a spring in motion. The difference between these two positions at time t is called the *displacement* and is denoted $x(t)$. Signs here will be choosen so that down is positive.

Assume exactly three forces act:

1. the restoring force of the spring, F_{spring},
2. an external force (driving the ceiling up and down, but may be 0), F_{ext},
3. a damping force (imagining the spring immersed in oil or that it is in fact a shock absorber on a car), F_{damp}.

In other words, the total force is given by

$$F_{total} = F_{spring} + F_{ext} + F_{damp}.$$

Physics tells us that the following are approximately true:

1. (Hooke's law [H]): $F_{spring} = -kx$, for some "spring constant" $k > 0$,
2. $F_{ext} = F(t)$, for some (possibly zero) function F,
3. $F_{damp} = -bv$, for some "damping constant" $b \geq 0$ (where v denotes velocity),
4. (Newton's 2nd law [N]): $F_{total} = ma$ (where a denotes acceleration).

Putting this all together, we obtain $mx'' = ma = -kx + F(t) - bv = -kx + F(t) - bx'$, or

$$\boxed{mx'' + bx' + kx = F(t).}$$

This is the **spring equation**. When $b = F(t) = 0$ this is also called the equation for simple harmonic motion.

Consider again first two figures above: (a) a spring at rest, without mass attached and (b) a spring at rest, with mass attached. The mass in the second figure is at rest, so the gravitational force on the mass, mg, is balanced by the restoring force of the spring: $mg = ks$, where s is the stretch.In particular, the spring constant can be computed from the stretch:

$$\boxed{k = \frac{mg}{s}.}$$

Example:

A spring at rest is suspended from the ceiling without mass. A 2 kg weight is then attached to this spring, stretching it 9.8 cm. From a position 2/3 m above equilibrium the weight is give a downward velocity of 5 m/s.

(a) Find the equation of motion.

(b) What is the amplitude and period of motion?

(c) At what time does the mass first cross equilibrium?

(d) At what time is the mass first exactly 1/2 m below equilibrium?

We shall solve this problem using SAGE below. Note $m = 2$, $b = F(t) = 0$ (since no damping or external force is even mentioned), and $k = mg/s = 2 \cdot 9.8/(0.098) = 200$. Therefore, the DE is $2x'' + 200x = 0$. This has general solution $x(t) = c_1 \cos(10t) + c_2 \sin(10t)$. The constants c_1 and c_2 can be computed from the initial conditions $x(0) = -2/3$ (down is positive, up is negative), $x'(0) = 5$.

Using SAGE, the displacement can be computed as follows:

SAGE

```
sage: t = var('t')
sage: x = function('x', t)
sage: m = var('m')
sage: b = var('b')
sage: k = var('k')
sage: F = var('F')
sage: de = lambda y: m*diff(y,t,t) + b*diff(y,t) + k*y - F
```

```
sage: de(x(t))
-F + m*diff(x(t), t, 2) + b*diff(x(t), t, 1) + k*x(t)
sage: m = 2; b = 0; k = 2*9.8/(0.098); F = 0
sage: de(x(t))
2*diff(x(t), t, 2) + 200.000000000000*x(t)
sage: desolve(de(x(t)),[x,t])
'%k1*sin(10*t)+%k2*cos(10*t)'
sage: desolve_laplace(de(x(t)),["t","x"],[0,-2/3,5])
'sin(10*t)/2-2*cos(10*t)/3'
```

Now we write this in the more compact and useful form $A\sin(\omega t + \phi)$ using the formulas

$$c_1\cos(\omega t) + c_2\sin(\omega t) = A\sin(\omega t + \phi),$$

where $A = \sqrt{c_1^2 + c_2^2}$ denotes the *amplitude* and $\phi = 2\arctan(\frac{-2/3}{1/2+A})$.

SAGE

```
sage: A = sqrt((-2/3)^2+(1/2)^2)
sage: A
5/6
sage: phi = 2*atan((-2/3)/(1/2 + A))
sage: phi
-2*atan(1/2)
sage: RR(phi)
-0.927295218001612
sage: sol = lambda t: sin(10*t)/2-2*cos(10*t)/3
sage: sol2 = lambda t: A*sin(10*t + phi)
sage: P = plot(sol(t),0,2)
sage: show(P)
```

This is displayed below[3].

[3] You can also, if you want, type `show(plot(sol2(t),0,2))` to check that these two functions are indeed the same.



Figure 4: Plot of $2x'' + 200x = 0$, $x(0) = -2/3$, $x'(0) = 5$, for $0 < t < 2$.

Of course, the period is $2\pi/10 = \pi/5 \approx 0.628$.

To answer (c) and (d), we solve $x(t) = 0$ and $x(t) = 1/2$:

SAGE

```
sage: solve(A*sin(10*t + phi) == 0,t)
[t == atan(1/2)/5]
sage: RR(atan(1/2)/5)
0.0927295218001612
sage: solve(A*sin(10*t + phi) == 1/2,t)
[t == (asin(3/5) + 2*atan(1/2))/10]
sage: RR((asin(3/5) + 2*atan(1/2))/10)
0.157079632679490
```

In other words, $x(0.0927...) \approx 0$, $x(0.157...) \approx 1/2$.

Exercise: *Use the problem above.*

(a) At what time does the weight pass through the equilibrium position heading down for the 2nd time?

(b) At what time is the weight exactly 5/12 m below equilibrium and heading up?

References

[H] General wikipedia introduction to Hooke's Law `http://en.wikipedia.org/wiki/Hookes_law`

[H1] Wikipedia entry for Robert Hooke: `http://en.wikipedia.org/wiki/Robert_Hooke`

[H2] MacTutor entry for Hooke: `http://www-groups.dcs.st-and.ac.uk/%7Ehistory/Biographies/Hooke.html`

[N] Wikipedia entry for Newton's laws of motion (including Newton's 2nd law): `http://en.wikipedia.org/wiki/Newton%27s_laws_of_motion`

[S] Wikipedia entry for Simple harmonic motion: `http://en.wikipedia.org/wiki/Simple_harmonic_motion`